Computers & Geosciences Vol. 16, No. 3, pp. 385–393, 1990
Printed in Great Britain.

0098-3004/90 $3.00 + 0.00
Pergamon Press plc

SHORT NOTE

A FORTRAN PROGRAM TO CALCULATE SEISMIC ANISOTROPY FROM THE LATTICE PREFERRED ORIENTATION OF MINERALS

DAVID MAINPRICE
Laboratoire de Tectonophysique, U.S.T.L., Place Eugene Bataillon, 34060 Montpellier Cedex, France

(*Accepted* 4 *August* 1989)

INTRODUCTION

Seismic compressional wave anisotropy was observed first in the northeast Pacific where the velocity varies with azimuth (Raitt, 1963). Hess (1964) was the first to suggest that the lattice preferred orientation (LPO) of the minerals in the upper mantle were the cause of the anisotropy. Since that date many studies have confirmed the relationship between the seimic anisotropy of upper mantle rocks and their LPO (e.g. Christensen, 1966: Crosson and Lin, 1971; Baker and Carter, 1972; Pelsnich, Nicolas, and Stevenson, 1974; Christensen and Lundquist, 1982). Now it is well established that the LPO is the result of extensive plastic flow in the upper mantle (Nicolas and others, 1971; Baker and Carter, 1972) and hence there is a direct relationship between the kinematics of plastic flow, LPO, and seismic anisotropy (see Nicolas and Poirier, 1976; Mainprice and Nicolas, 1989).

More recently there has been considerable interest in the seismic anisotropy of the continental crust. In part this interest stems from the possibility of using seismic anisotropy, particuarly shear wave splitting (Crampin, 1981; Crampin, Evans, and Atkinson, 1984), as an indicator of dilatancy for earthquake prediction. Applications of this type have been programmed by McGonigle and Crampin (1982) for orthorombic or higher symmetry. Here I present another application, the contibution of LPO to the seismic anisotropy of seismic reflectors in the lower crust. Although there are several possible origins for such reflectors, their geometry strongly suggests that they are ductile shear zones. The composition of the lower crust is well constrained to be plagioclase feldspar-rich (< 60% by vol), with a well-developed LPO (Ji and Mainprice, 1988).

PROGRAM ALGORITHM

The seismic velocity anisotropy of a polycrystalline rock can be calculated if one knows the volume fraction F, density ρ, elastic constants C_{ij}, and the LPO of each mineral. The calculation can be made at any temperature and pressure if the appropriate elastic derivatives are known. Here I present a general analytical solution for triclinic mineral symmetry which is suitable particularly for microcomputers. The data required for the calculation, which serves as input for the program, are the elastic constants and density (see Simmons and Wang, 1971 for compilation), volume fraction of each mineral and the LPO. The volume fraction can be determined by the classical point counting technique or a computerized method (Allard and Sotin, 1989). The LPO is determined most conviently by measurement of individual grains with an optical polarizing microscope equipped with a universal stage (see Wenk, 1985 for review). For the current program the geographical coordinates of two of crystallographic reference directions ($X1$, $X3$ used to measure the elastic constants, see Nye, 1957) are used to define the crystal orientation in geographical coordinates. The third direction ($X2$) is calculated by the vector cross-product as these directions are orthogonal. The geographical coordinates (dip and azimuth) are converted to direction cosines which are elements of the crystal to geographic frame rotation matrix (R_{ij}). Now for each crystal (grain), the elastic frame (C_{pjrs}) defined by crystallographic directions of $X1$, $X2$, $X3$ to the geographic frame (C_{ijkl}) defined by the dip and azimuth of $X1$, $X2$, $X3$. The rotation is achieved by the standard four-rank tensor transformation (Nye, 1957, p. 133),

$$C_{ijkl} = R_{ip} R_{jg} R_{kr} R_{ls} C_{pgrs}.$$

The average elastic properties of the aggregate in geographical coordinates now can be calculated by summing the contribution of each mineral species (S) of volume fraction (F_s) consisting of N_s grains. There are several possible averaging schemes for elastic constants of which the Voigt, Reuss, and Voight–Reuss–Hill are the best known (Crosson and Lin, 1971; Baker and Carter, 1972). Previous comparatives studies (Crosson and Lin, 1971; Pelsnick, Nicolas, and Stevenson, 1974) between the anisotropy calculated from LPO and experimentally measured values show that the Voigt scheme consistently gives the best agreement. The Voigt average elastic stiffness matrix (C^{V}_{ijkl}) is given by

$$C^{V}_{ijkl} = \sum_{1}^{s} F_s \sum_{1}^{N_s} C_{ijkl}/N_s.$$

The final step in the calculation is to evaluate the seismic velocities for each direction of interest (X_i) over a geographic hemisphere. It is traditional to present this as solution of the Christoffel equation (e.g. Crosson and Lin, 1971; Peselnick, Nicolas, and Stevenson, 1974) defined as

$$\det| T_{ik} - \delta_{ik}\rho V^2 | = 0$$

where δ_{ik} is the Kronecker delta, V is one of the three seismic velocities and T_{ik} the Christoffel stiffness. In practice a general solution (with no symmetry restrictions) is achieved most easily by solving for the eigenvalues of the asymmetric Christoffel stiffness matrix (T_{ik}) which is defined by

$$T_{ik} = C_{ijkl}X_jX_l$$

where X_jX_l are the direction cosines of the geographical direction of interest. The eigenvalues of an asymmetric matrix are difficult to obtain by standard methods, whereas a simple analytical solution is available for symmetric matrices. The multiplication of asymmetric matrix (T_{ik}) by its transpose (T_{jk}) yields a symmetric matrix (TT_{ij}) whose eigenvalues are the square of the asymmetric matrix (T_{ik}), viz.

$$TT_{ij} = T_{ik} \cdot T_{jk}.$$

The eigenvalues (E_i) of the symmetric matrix TT_{ij} are determined analytically by solving the cubic equation in seismic velocity whose roots are all real and unequal by a simplified trigonometrical solution.

$$v^3 - av^2 + bv - c = 0$$

where

$$a = TT_{ii},\ b = 1/2\,(TT_{ii}TT_{jj} - TT_{ij}TT_{ij}),\ c = \det|TT_{ij}|.$$

Hence the three seismic velocities are simply given by

$$V_i = E_i^{1/4}\rho^{-1/2}.$$

To simplify the identification of the quasicompression (qP) and two quasishear velocities (qS_1, qS_2), the eigenvalues are sorted into decending order of magnitude, and knowing that velocity $qP > qS_1$ we can store conveniently the calculated velocities. The calculation is repeated for a grid of orientations with a 6° spacing to cover an entire hemisphere. The three velocities qP, qS_1, and qS_2 are stored on diskette for later use by a contouring routine, in this application a BASIC translation of the routine described by Kalami and Von Frese (1980) is used.

APPLICATION

The program has been used to calculate the seismic anisotropy of plagioclase feldspar tectonites (Ji and Mainprice, 1988) from amphibole facies metamorphic terrains. The geophysical interest in understanding the physical properties of lower crustal tectonites (amphibolite and granulite facies) particularly is acute with the current problems of interpretation of the seismic sections generated by the regional deep-seismic reflection profiling of the continental crust [e.g. ECORS (France), COCORP (U.S.A.), BIRPS (U.K.), DEKORP (Germany), and ACORP (Australia)]. A survey of the LPOs of lower crustal rocks and their seismic anisotropy has been undertaken. As an example the seismic anisotropy of qP, is taken in plagioclase feldspar single crystal (Fig. 1A) and a plagioclase tectonite from Norway (Fig. 1B). Firstly, note that although the symmetry of plagioclase is triclinic, the symmetry of its elastic properties and hence its seismic anisotropy is monoclinic with the (010) plane acting as a mirror plane. The monoclinic symmetry is an artifact of the mathematical proceedure used to calculate the elastic constants from the experimentally observed seismic velocites (Ryzhova, 1964). The pseudomonoclinic symmetry infact is more representative of the properties of deformed plagioclase grains as they are twinned extensively on the (010) plane (Albite twinning). Secondly, the strong LPO of plagioclase tectonites results in strong seismic anisotropies with the anisotropy coefficient (V_{max} - V_{min}/V_{max} × 100%) of between 14 and 17% for qP velocities. The maximum qP velocity is normal to the



Figure 1. A, qP velocities of single crystal of plagioclase (anorthite content 53%) in km/s; B, qP velocities of polycrystalline plagioclase tectonite from southern Norway. Average plagioclase composition is 45% anorthite.

foliation plane (Z) and the minimum is normal to the lineation in the foliation plane(Y). Such strong anisotropies will result in variations in qP with azimuth for horizontally foliated regions, such as the crustal laminations seen in northern France (Bois and others, 1987) and good seismic reflectors in dipping ductile shear zones (Jones and Nur, 1982).

Acknowledgments—This research was supported by the CNRS INSU "ATP-Accompagnement ECORS 87" and UA 732. Thanks go to Shaocheng for the example LPO data, Yvan Montardi for helpful discussions and Prof. N. Christensen for reviewing the manuscript.

REFERENCES

Allard, B., and Sotin C., 1989, Determination of percentages of different mineral phases in granular rocks using image analysis on a microcomputer: Computers & Geosciences, v. 15, no. 2, p. 441–448.

Baker, D.W., and Carter, N.L., 1972, Seismic velocity anisotropy calculated from ultramafic minerals and aggregates, *in* Flow and fracture of rocks: Am. Geophys. Union Monograph 16, p. 157–166.

Bois, C., Cazes, M., Hirn, A., Matte, P., Mascle, A., Montadert, L., and Pinet, B., 1987, Crustal laminations in deep seismic profiles in France and neighbouring areas: Geophys. Jour. Roy. Astr. Soc., v. 89, no. 1, p. 297–304.

Christensen, N.I., 1966, Elasticity of ultrabasic rocks: Jour. Geophys. Res., v. 71, no. 5, p. 5921–5931.

Christensen, N.I., and Lundquist, S.M., 1982, Pyroxene orientation within the upper mantle: Geol. Soc. America Bull., v. 93, no. 4 p. 279–288.

Crampin, S., 1981, A review of wave motion in anisotropic and cracked elastic-media: Wave Motion, v. 3, no. 2, p. 343–391.

Crampin, S., Evans, R., and Atkinson, B.K., 1984, Earthquake prediction: a new physical basis: Geophys. Jour. Roy. Astr. Soc., v. 76, no. 1, p. 147–156.

Crosson, R.S., and Lin, J.W., 1971, Voigt and Reuss prediction of anisotropic elasticity of olivine: Jour. Geophysical Res., v. 76, no. 4, p. 570–578.

Hess, H.H., 1964, Seismic anisotropy of the uppermost mantle under oceans: Nature, v. 204, no. 2, p. 629–631.

Ji, S., and Mainprice, D., 1988, Natural deformation fabrics of plagioclase: implications for slip systems and seismic anisotropy: Tectonophysics, v. 147, no. 1, p. 145–163.

Jones, T., and Nur, A., 1982, Seismic velocity and anisotropy in mylonites and the reflectivity of deep crustal fault zones: Geology, v. 10, no. 5, p. 260–263.

Kalkani, E.C., and Von Frese, R.R.B., 1980, Computer construction of equal-angle fabric diagrams and program comparisons: Computers & Geosciences, v. 6, no. 3, p. 279–288.

McGonigle, R., and Crampin S., 1982, A FORTRAN program to evaluate the phase and group-velocity surface in an anisotropic solid: Computers & Geosciences, v. 8, no. 2, p. 221–226.

Mainprice, D., and Nicolas A., 1989, Development of shape and lattice preferred orientations: application to the seismic anisotropy of the lower crust: Jour. Struct. Geology, v. 11, no. 1/2, p. 175–189.

Nicolas, A., Bouchez, J.L., Boudier, F., and Mercier J.C., 1971, Textures, structures and fabrics due to solid state flow in some European lherzolities: Tectonophysics, v. 12, no. 1, p. 55–86.

Nicolas, A., and Poirier, J.P., 1976, Crystalline plasticity and solid state flow in metamorphic rocks: John Wiley & Sons, New York, 444 p.

Nye, J.F., 1957, Physical properties of crystals: Clarendon Press, Oxford, 322 p.

Peselnick, L., Nicolas, A., and Stevenson, P.R., 1974, Velocity anisotropy in a mantle peridotite from the Ivrea zone: Application to upper mantle anisotropy: Jour. Geophys. Res., v. 79, no. 8, p. 1175–1182.

Raitt, R.W., 1963, Seismic refraction studies of the Mendicina fracture zone: Rept. MPL-U-23/63: Marine Phys. Lab. Scripps Institute of Oceanography, Univ. California, San Diego, 12 p.

Ryzhova, T.V., 1964, Elastic properties of plagioclase: Akad. Nauk. USSR, Bull. Geophys. Ser., 7, p. 633–635.

Simmons, G., and Wang, H., 1971, Single crystal elastic constants and calculated aggregate properties: a handbook (2nd ed.): The MIT Press, Cambridge, Massachusetts, 354 p.

Wenk, H.R., 1985, Measurement of pole figures, *in* Wenk H.R. ed., Preferred orientation in deformed metals and rocks: an introduction to modern texture analysis: Academic Press, Orlando, 610 p.

APPENDIX

Program Listing

```
C
C Seismic anisotropy from lattice preferred orientation
C
C                   DAVID MAINPRICE
C               LABORATOIRE DE TECTONOPHYSIQUE
C               U.S.T.L.,PLACE E.BATAILLON,
C               34060 MONTPELLIER,FRANCE.
C The orientaion of the elastic constants cartesian
C  reference frame axes X1 and X3 are to be given
C  in geographic coordinates (azimuth,dip).
C Elastic constant right-handed reference frame
C  conventions for axes X1,X2,X3
C  cubic,tetragonal,othorhombic:A,B,C
C  hexagonal,trigonal:A1,(C x A1),C
C  monoclinic:A,B,C* (alkali feldspar) two conventions X2=B or X3=B
C                A*,B,C (diopside,augite,hornblende)
C  triclinic :A,B,C* (playiclase) no general convention
C
```

```
C*********************************************************
      CHARACTER FILE1*15,FILE2*15,FILE3*15,TITLE*40,REF(2)*40,ANS*1
      CHARACTER MINERAL*15
      DIMENSION X1(3),X2(3),X3(3),XI(3),R(3,3),A(3),E(3)
      DIMENSION V(3),GV(3),P(3,3),G(3,3),GAZ(3),GDIP(3)
      DIMENSION CSUM(6,6),CM(6,6),C(6,6),EC(6,6)
      COMMON/SUBS/L1(6),L2(6),IJKL(3,3)
      L1(1)=1
      L1(2)=2
      L1(3)=3
      L1(4)=2
      L1(5)=3
      L1(6)=1
      L2(1)=1
      L2(2)=2
      L2(3)=3
      L2(4)=3
      L2(5)=1
      L2(6)=2
      IJKL(1,1)=1
      IJKL(1,2)=6
      IJKL(1,3)=5
      IJKL(2,1)=6
      IJKL(2,2)=2
      IJKL(2,3)=4
      IJKL(3,1)=5
      IJKL(3,2)=4
      IJKL(3,3)=3



C*********************************************************
C     INPUT SECTION
C*********************************************************
1000  WRITE(*,'('' PROGRAM ANS3'')')
      WRITE(*,
     *'('' Seismic anisotropy from LPO '')')
      WRITE(*,'('' TITLE ?'')')
      READ(*,'(A40)') TITLE
      WRITE(*,'('' FILE NAME FOR OUTPUT CONTOUR DATA ?'')')
      READ(*,'(A15)') FILE2
      OPEN(7,FILE=FILE2,STATUS='NEW',ACCESS='SEQUENTIAL'
     1, FORM='FORMATTED')
      WRITE(*,'('' Hemisphere Convention: '')')
      WRITE(*,'('' +1=dip +VE upper hemisphere '')')
      WRITE(*,'('' -1=dip +VE lower hemisphere '')')
      READ(*,*) IHEMI
      WRITE(*,'('' Hemisphere Convention For Contouring: '')')
      WRITE(*,'('' +1= dip +VE hemisphere '')')
      WRITE(*,'('' -1= dip -VE hemisphere '')')
      READ(*,*) ICONT
      DO 10 I=1,6
      DO 10 J=1,6
10    C(I,J)=0.0
      DROCK=0.0
      VOLSUM=0.0
C**************************************************
C
C     MINERAL LOOP
C
C**************************************************
      WRITE(*,'('' HOW MANY MINERALS ?'')')
      READ(*,*) NMIN
      DO 20 MIN=1,NMIN
      WRITE(*,15) MIN
15    FORMAT(1X,'NAME OF MINERAL',I2,' ?')
```

```
      READ(*,'(A15)') MINERAL
      WRITE(*,'('' Volume Faction Of Aggregate in Range 0 to 1'')')
      READ(*,*) VOL
      VOLSUM=VOLSUM+VOL
      IF (MIN.NE.NMIN)GOTO 24
      IF (VOLSUM.EQ.1.00)GOTO 24
      WRITE(*,'('' *Warning* Volume Fractions Do Not Total To 1.00'')')
      WRITE(*,23) VOLSUM
23    FORMAT(1X,'Volume Fraction Total =',F6.4)
24    WRITE(*,25) MINERAL
25    FORMAT(1X,'Name of File Containing Elastic Constant Data for '
     *,A15,' ?')
      READ(*,'(A15)') FILE1
      OPEN(9,FILE=FILE1,STATUS='OLD',ACCESS='SEQUENTIAL'
     *, FORM='FORMATTED')
C READ ELASTIC CONSTANTS FROM FILE 9
      READ(9,30)REF,A,E,INDEX,EC
30    FORMAT(A40/A40/6F8.4,I1/6F8.4/6F8.4/6F8.4/6F8.4/6F8.4/6F8.4)
C CLOSE OUTPUT FILE 9
      CLOSE(9,STATUS='KEEP')
      WRITE(*,'('' Density (g/cm3)'')')
      READ(*,*) RHO
      WRITE(*,35) MINERAL
35    FORMAT(1X,'Name of File Containing Azimuth/Dip Data for ',A15,'?')
      READ(*,'(A15)') FILE3
      OPEN(8,FILE=FILE3,STATUS='OLD',ACCESS='SEQUENTIAL'
     *, FORM='FORMATTED')
      DO 40 I=1,6
      DO 40 J=1,6
      CM(I,J)=0.0
40    CSUM(I,J)=0.0
      WRITE(*,'('' SWITCH ON PRINTER ALT-P AND TYPE 1'')')
      READ(*,*) Q
      REWIND 8
      NGRAIN=0
      IFLAG=0
C**************************************************
C

C     GRAIN LOOP

C
C**************************************************
C     READ AZ/DIP
50    READ(8,*,END=65) AZM1,DIP1,AZM3,DIP3
      NGRAIN=NGRAIN+1
C**************************************************
C     CONVERT FROM GEOGRAPHIC COORDINATES TO DIRECTION COSINES
C**************************************************
      CALL CART(DIP1,AZM1,IHEMI,X1)
      CALL CART(DIP3,AZM3,IHEMI,X3)
C AXIS-2= VECTOR CROSS-PRODUCT 3 X 1
      CALL XPROD(X3,X1,X2)
C REPEAT TO ROUND-OFF ERRORS
      CALL XPROD(X1,X2,X3)
      CALL XPROD(X2,X3,X1)
      CALL XPROD(X3,X1,X2)
      CALL ANGLE(X1,X3,VALUE)
      TEST=ABS(90.0-VALUE)
      IF(TEST.LT.3.0)GOTO 55
C ERROR X1 AND X3 NOT PERPENDICULAR WITHIN 3 DEGREES MARGIN
      IFLAG=IFLAG+1
      WRITE(*,200) NGRAIN,VALUE
      WRITE(*,210) DIP1,AZM1
      WRITE(*,220) DIP3,AZM3
C**************************************************
C     ROTATE ELASTIC CONSTANTS FROM CRYSTAL FRAME TO GEOGRAPHIC
C**************************************************
55    CALL ROTEC(X1,X2,X3,R,EC,CM)
```

```
      DO 60 I=1,6
      DO 60 J=1,6
60    CSUM(I,J)=CSUM(I,J)+CM(I,J)
      GOTO 50
C************************************************
C     CALCULATE AVERAGE ELASTIC CONSTANTS FOR MINERAL
C************************************************
65    DO 70 I=1,6
        DO 70 J=1,6
70    CM(I,J)=CSUM(I,J)/FLOAT(NGRAIN)
        WRITE(*,75) FILE3,IFLAG
75    FORMAT(5X,'ALL DATA READ FROM ',A15,' ERROR COUNT = ',I2)
        WRITE(*,76)
76    FORMAT(//5X,'CONTINUE PROCESSING THIS MINERAL FILE (Y/N) ?'
     */10X,'SWITCH-OFF PRINTER ALT-N')
      READ(*,'(A1)') ANS
      IF(ANS.EQ.'N')GOTO 20
      WRITE(*,202) TITLE
      WRITE(*,300)
      DO 310 J=1,6
310   WRITE(*,320)(EC(I,J),I=1,6)
      WRITE(*,'(A40,70X,A40)')REF(1),REF(2)
      WRITE(*,330) RHO
      WRITE(*,340) MINERAL
340   FORMAT(1X,'CALCULATED AVERAGE ELASTIC STIFFNESS FOR ',A15)
      DO 350 J=1,6
350   WRITE(*,320)  (CM(I,J),I=1,6)
C SUM FOR AGGREGATE
      WRITE(*,360) VOL
360   FORMAT(/1X,'VOLUME FRACTION :',F6.4)
      DO 370 I=1,6
      DO 370 J=1,6
370   C(I,J)=C(I,J)+VOL*CM(I,J)
C     ROCK DENSITY
      DROCK=DROCK+VOL*RHO
C***********************************
C
C     END OF MINERAL LOOP
C

C***********************************
20    CONTINUE
      WRITE(*,380) TITLE
380   FORMAT(1X,'CALCULATED AVERAGE ELASTIC STIFFNESS FOR '
     *'AGGREGATE'/A40)
      DO 390 J=1,6
390   WRITE(*,320)  (C(I,J),I=1,6)
      WRITE(*,395) DROCK
395   FORMAT(/1X,'AGGREGATE DENSITY =',F7.4,' g/cm3')
C************************************************
C     PLANE WAVE CALCULATION OVER WHOLE GEOGRAPHIC HEMISPHERE
C     USING A GRID SPACING OF 6 DEGREES

C************************************************
      WRITE(7,2027) TITLE
      DO 110 I=1,16
      DO 110 J=1,61
      DIPI=FLOAT(ICONT)*FLOAT(90-((I-1)*6))
      AZMI=FLOAT(90*(J-1)/15)
C     CONVERT GEOGRAPHIC COORDINATES TO DIRECTION COSINES
      CALL CART(DIPI,AZMI,IHEMI,XI)
C     CALCULATE PHASE VELOCITES
      CALL VELO(XI,DROCK,C,V)
C     WRITE TO FILE 7
      WRITE(7,240) V(1),V(2),V(3)
      WRITE(*,230) DIPI,AZMI,V(1)
110   CONTINUE
C***********************
C     END LOOP
C***********************
```

```
C CLOSE OUTPUT FILE 7
      CLOSE(7,STATUS='KEEP')
C
      GOTO 1000
C FORMATS
202   FORMAT(//2X,'PLANE WAVE ANALYSIS',8X,A40)
2027  FORMAT(2X,A40)
200   FORMAT(//5X,'* X1 AND X3 NOT PERPENDICULAR IN GRAIN ',I3,' *'
     */8X,'ANGLE BETWEEN X1 AND X3 =',F6.2,' DEGREES')
210   FORMAT(8X,'AXIS-1: Dip= ',F5.1,' Azimuth=',F5.1,
     *' degrees')
220   FORMAT(8X,'AXIS-3: Dip= ',F5.1,' Azimuth=',F5.1,
     *' degrees')
230   FORMAT(/1X,'Dip=',F5.1,' Azimuth=',F5.1,' Vp =',F5.1)
240   FORMAT(2(F6.2,','),F6.2)
300   FORMAT(/,1X,'ELASTIC STIFFNESS MATRIX (Mbars)')
320   FORMAT(/1X,6F9.4)
330   FORMAT(/1X,'Density=',F7.4,' g/cm3')
      END



      SUBROUTINE VELO(X,RHO,C,V)
C PHASE-VELOCITY SURFACES IN AN ANISOTROPTIC MEDIUM
C     X(3) - DIRECTION OF INTEREST
C     RHO - DENSITY
C     V - PHASE VELOCITIES (1,2,3= P,S,SS)
      DIMENSION V(3),C(6,6),T(3,3),X(3),EVAL(3),TT(3,3)
      COMMON/SUBS/L1(6),L2(6),IJKL(3,3)
C FORM ASYMMETRIC MATRIX Tik=Cijkl*Xj*Xl
      DO 10 I=1,3
      DO 10 K=1,3
      T(I,K)=0.0
      DO 20 J=1,3
      DO 20 L=1,3
      M=IJKL(I,J)
      N=IJKL(K,L)
      T(I,K)=T(I,K)+C(M,N)*X(J)*X(L)
20    CONTINUE
10    CONTINUE
C FORM SYMMETRICAL MATRIX TTij=Tik*Tjk
      DO 30 I=1,3
      DO 30 J=1,3
      TT(I,J)=0.0
      DO 40 K=1,3
40    TT(I,J)=TT(I,J)+T(I,K)*T(J,K)
30    CONTINUE
C DETERMINE THE EIGENVALUES OF SYMMETRIC TTij
      CALL EIGEN3(TT,EVAL)
C EIGENVALUES OF TTij ARE THE SQUARE OF Tij
C CALCULATE PHASE VELOCITY
      DO 50 I=1,3
50    V(I)=10.0*SQRT(SQRT(EVAL(I))/RHO)
      RETURN
      END

      SUBROUTINE EIGEN3(X,V)
C EIGENVALUES OF SYMMETRIC X(3,3) IN V(3)
      DIMENSION X(3,3),V(3),T(3)
      PI=4.0*ATAN(1.0)
      A=-(X(1,1)+X(2,2)+X(3,3))
      B=X(1,1)*X(2,2)-X(1,2)*X(1,2)+X(1,1)*X(3,3)-X(1,3)*X(1,3)
     1+X(2,2)*X(3,3)-X(2,3)*X(2,3)
      C=-(X(1,1)*X(2,2)*X(3,3)+2*X(1,2)*X(1,3)*X(2,3)
     1-X(1,1)*X(2,3)*X(2,3)-X(2,2)*X(1,3)*X(1,3)-X(3,3)*X(1,2)*X(1,2))
      Q=(3.0*B-A*A)/9.0
      R=(9.0*A*B-27.0*C-2.0*A*A*A)/54.0
      Y=-R/SQRT(-Q*Q*Q)
      IF(Y.LT.-1.0) Y=-1.0
```

```
      IF(Y.GT.1.0) Y=1.0
      THETA=ACOS(Y)
      Y1=2.0*SQRT(-Q)
      T(1)=THETA/3.0
      T(2)=T(1)+2.0*PI/3.0
      T(3)=T(1)+4.0*PI/3.0
C SORT EIGENVALUES INTO DECENDING ORDER
      DO 10 I=1,3
      CS=COS(T(I))
      IF(T(I).GT.PI)CS=-COS(T(I)-PI)
10    V(I)=-(Y1*CS+A/3.0)
      DO 20 I=1,2
      N=I+1
      DO 20 IN=N,3
      IF(V(IN).LE.V(I))GOTO 20
      Y=V(I)
      V(I)=V(IN)
      V(IN)=Y
20    CONTINUE
      RETURN
      END

      SUBROUTINE ROTEC(X1,X2,X3,R,EC,C)
C SET-UP TRANSFORMATION MATRIX CRYSTAL->SPACIAL
C ROTATE ELASTIC CONSTANT MATRIX (EC) FROM CRYSTAL
C TO SPACIAL FRAME (C)
      DIMENSION X1(3),X2(3),X3(3),R(3,3),EC(6,6),C(6,6),D(6,6)
      COMMON/SUBS/L1(6),L2(6),IJKL(3,3)
C SET-UP TRANSFORMATION (DIRECTION COSINE) MATRIX CRYSTAL->SPACIAL
C THREE CRYSTALLOGRAPHIC AXES IN SPACIAL CARTESIAN COORDINATES
C AXIS-1:AXIS-3:AXIS-2=3 X 1
C FILL ROTATION MATRIX WITH AXES BY COLUMNS
      DO 10 I=1,3
      R(I,1)=X1(I)
      R(I,2)=X2(I)
10    R(I,3)=X3(I)
C NORMALIZE TO DIRECTION COSINES
      DO 20 J=1,3
      XM=SQRT(R(1,J)**2+R(2,J)**2+R(3,J)**2)
      DO 30 I=1,3
30    R(I,J)=R(I,J)/XM
20    CONTINUE
      DO 40 I=1,6
      DO 40 J=1,6
40    D(J,I)=EC(J,I)
C ROTATE ELASTIC CONSTANTS FORM CRYSTAL TO SPACIAL COORDINATES
C Cijkl=Rip*Rjq*Rkr*Rls*Cpqrs
      DO 34 M=1,6
      I=L1(M)
      J=L2(M)
      DO 34 N=1,M
      K=L1(N)
      L=L2(N)
      X=0.0
      DO 33 LP=1,3
      Y=0.0
      DO 32 LQ=1,3
      LT=IJKL(LP,LQ)
32    Y=Y+R(J,LQ)*
     1  (R(K,1)*(R(L,1)*D(LT,1)+R(L,2)*D(LT,6)+R(L,3)*D(LT,5))
     2  +R(K,2)*(R(L,1)*D(LT,6)+R(L,2)*D(LT,2)+R(L,3)*D(LT,4))
     3  +R(K,3)*(R(L,1)*D(LT,5)+R(L,2)*D(LT,4)+R(L,3)*D(LT,3)))
33    X=X+R(I,LP)*Y
      C(M,N)=X
34    C(N,M)=X
      RETURN
      END
      SUBROUTINE CART(DIP,AZM,IHEMI,X)
      DIMENSION X(3)
```

```
      DATA CON/0.0174532925/
C CONVERT FROM SPHERICAL TO CARTESIAN CO-ORDINATES
C SOUTH 1,0,0  EAST 0,1,0 UP 0,0,1 DOWN 0,0,-1
C IHEMI=-1 LOWER HEMISPHERE +VE DIP
C IHEMI=+1 UPPER HEMISPHERE +VE DIP
      CAZ=COS(AZM*CON)
      SAZ=SIN(AZM*CON)
      DIPH=FLOAT(IHEMI)*DIP
      CDIP=COS(DIPH*CON)
      SDIP=SIN(DIPH*CON)
      X(1)=-CDIP*CAZ
      X(2)=CDIP*SAZ
      X(3)=SDIP
C NORMALISE TO DIRECTION COSINES
      R=SQRT(X(1)*X(1)+X(2)*X(2)+X(3)*X(3))
      DO 10 I=1,3
10    X(I)=X(I)/R
      RETURN
      END

      SUBROUTINE ANGLE(A,B,VALUE)
C ANGLE BETWEEN DIRECTION COSINES A,B
      DIMENSION A(3),B(3)
      DOT=A(1)*B(1)+A(2)*B(2)+A(3)*B(3)
      IF(DOT.GT.1.0)DOT=1.0
      IF(DOT.LT.-1.0)DOT=-1.0
      VALUE=ACOS(DOT)*57.29577951
      RETURN
      END

      SUBROUTINE XPROD(A,B,C)
C CROSS VECTOR PRODUCT C=A X B
      DIMENSION A(3),B(3),C(3),D(3)
      D(1)=A(2)*B(3)-A(3)*B(2)
      D(2)=A(3)*B(1)-A(1)*B(3)
      D(3)=A(1)*B(2)-A(2)*B(1)
      DO 10 I=1,3
10    C(I)=D(I)
      RETURN
      END
```